/* Copyright (C) 2014-2016 Dan Chapman <dpniel@ubuntu.com>

   This file is part of the dekko

   This program is free software; you can redistribute it and/or
   modify it under the terms of the GNU General Public License as
   published by the Free Software Foundation; either version 2 of
   the License or (at your option) version 3 or any later version
   accepted by the membership of KDE e.V. (or its successor approved
   by the membership of KDE e.V.), which shall act as a proxy
   defined in Section 14 of version 3 of the license.

   This program is distributed in the hope that it will be useful,
   but WITHOUT ANY WARRANTY; without even the implied warranty of
   MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
   GNU General Public License for more details.

   You should have received a copy of the GNU General Public License
   along with this program.  If not, see <http://www.gnu.org/licenses/>.
*/
import QtQuick 2.4
import Ubuntu.Components 1.3
import Dekko.Theme 0.2

ListItem {
    //FIXME: Some reason we have to use the qtObject here??? the rolenames should be mapped correctly
    property var attachment
    height: delegateLayout.height
    signal openAttachment(var url)

    onClicked: {
        delegateLayout.summary.text = ""
        if (attachment.partUrl) {
            openAttachment(attachment.partUrl)
        } else {
            attachment.fetch()
        }
    }

    ListItemLayout {
        id: delegateLayout
        title.text: attachment.fileName
        title.fontSize: "x-small"
        title.elide: Text.ElideRight
        title.wrapMode: Text.NoWrap
        title.font.weight: Font.normal
        subtitle.text: attachment.mimeType + ", " + attachment.size
        subtitle.fontSize: "medium"
        subtitle.elide: Text.ElideRight
        subtitle.wrapMode: Text.NoWrap
        subtitle.font.weight: Font.Normal
        summary.visible: summary.text !== ""
        summary.text: ""
        summary.fontSize: "medium"
        summary.color: "red"
        Icon {
            name: attachment.iconName
            color: "#888888"
            height: units.gu(4); width: height
            SlotsLayout.position: SlotsLayout.Leading
        }

        ActivityIndicator {
            visible: running
            running: attachment.fetchInProgress
        }
    }

    Connections {
        target: attachment
        onDownloadProgressChanged: console.log("Download progress: ", downloadProgress)
        onError: {
            console.log("Download error: ", errorString)
            delegateLayout.summary.text = errorString
        }
        onPartUrlChanged: {
            console.log("Download URL: ", partUrl)
            if (partUrl) {
                openAttachment(partUrl)
            }
        }
    }

    Component.onDestruction: {
        if (attachment.partUrl) {
            // clean up any temporary files
            attachment.cleanup()
        }
    }
}
